

UCloud Provider

NOTE: This guide requires an available UCloud account or sub-account with project to create resources.

The UCloud provider is used to interact with the resources supported by UCloud. The provider needs to be configured with the proper credentials before it can be used.

Use the navigation to the left to read about the available resources.

Example Usage

```
# Configure the UCloud Provider
provider "ucloud" {
  public_key = "${var.ucloud_public_key}"
  private_key = "${var.ucloud_private_key}"
  project_id = "${var.ucloud_project_id}"
  region     = "cn-sh2"
}

# Query availability zone
data "ucloud_zones" "default" {
}

# Query image
data "ucloud_images" "default" {
  availability_zone = "${data.ucloud_zones.default.zones.0.id}"
  os_type           = "Linux"
}

# Create security group
resource "ucloud_security_group" "default" {
  name = "tf-example-eip"
  tag  = "tf-example"

  rules {
    port_range = "80"
    protocol   = "tcp"
    cidr_block = "192.168.0.0/16"
    policy     = "ACCEPT"
  }
}

# Create a web server
resource "ucloud_instance" "web" {
  instance_type      = "n-standard-1"
  availability_zone   = "${data.ucloud_zones.default.zones.0.id}"
  image_id            = "${data.ucloud_images.default.images.0.id}"

  root_password      = "wA1234567"
  security_group      = "${ucloud_security_group.default.id}"

  name                = "tf-example-eip"
  tag                  = "tf-example"
}
```

Authentication

The UCloud provider offers a flexible means of providing credentials for authentication. The following methods are supported, in this order, and explained below:

- Static credentials
- Environment variables

Static credentials

Static credentials can be provided by adding an `public_key` and `private_key` in-line in the UCloud provider block:

Usage:

```
provider "ucloud" {  
  public_key = "your_public_key"  
  private_key = "your_private_key"  
  project_id = "your_project_id"  
  region     = "cn-sh2"  
}
```

Environment variables

You can provide your credentials via `UCLLOUD_PUBLIC_KEY` and `UCLLOUD_PRIVATE_KEY` environment variables, representing your UCloud public key and private key respectively. `UCLLOUD_REGION` and `UCLLOUD_PROJECT_ID` are also used, if applicable:

```
provider "ucloud" {}
```

Usage:

```
$ export UCLLOUD_PUBLIC_KEY="your_public_key"  
$ export UCLLOUD_PRIVATE_KEY="your_private_key"  
$ export UCLLOUD_REGION="cn-sh2"  
$ export UCLLOUD_PROJECT_ID="org-xxx"  
  
$ terraform plan
```

Argument Reference

In addition to generic provider arguments (<https://www.terraform.io/docs/configuration/providers.html>) (e.g. `alias` and `version`), the following arguments are supported in the UCloud provider block:

- `public_key` - (Required) This is the UCloud public key. It must be provided, but it can also be sourced from the `UCLLOUD_PUBLIC_KEY` environment variable.
- `private_key` - (Required) This is the UCloud private key. It must be provided, but it can also be sourced from the `UCLLOUD_PRIVATE_KEY` environment variable.

- `region` - (Required) This is the UCloud region. It must be provided, but it can also be sourced from the `ULOUD_REGION` environment variables.
- `project_id` - (Required) This is the UCloud project id. It must be provided, but it can also be sourced from the `ULOUD_PROJECT_ID` environment variables.
- `max_retries` - (Optional) This is the max retry attempts number. Default max retry attempts number is 0.
- `insecure` - (Optional) This is a switch to disable/enable https. (Default: `false`, means enable https).

Testing

Credentials must be provided via the `ULOUD_PUBLIC_KEY`, `ULOUD_PRIVATE_KEY`, `ULOUD_PROJECT_ID` environment variables in order to run acceptance tests.

ucloud_eips

This data source provides a list of EIP resources (Elastic IP address) according to their EIP ID.

Example Usage

```
data "ucloud_eips" "example" {}

output "first" {
  value = "${data.ucloud_eips.example.eips.0.ip_set.0.ip}"
}
```

Argument Reference

The following arguments are supported:

- `ids` - (Optional) The IDs of Elastic IP, all the EIPs belong to this region will be retrieved if the ID is "".
- `output_file` - (Optional) File name where to save data source results (after running `terraform plan`).

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `eips` - eips is a nested type which documented below.
- `total_count` - Total number of Elastic IP that satisfy the condition.

The attribute (`eips`) support the following:

- `bandwidth` - Maximum bandwidth to the elastic public network, measured in Mbps.
- `ip_set` - It is a nested type which documented below.
- `create_time` - The time of creation for Elastic IP, formatted in RFC3339 time string.
- `expire_time` - The expiration time for Elastic IP, formatted in RFC3339 time string.
- `charge_mode` - Elastic IP charge mode. Possible values are: `traffic as pay by traffic`, `bandwidth as pay by bandwidth`.
- `charge_type` - Elastic IP Charge type. Possible values are: `year as pay by year`, `month as pay by month`, `dynamic as pay by hour`.
- `name` - The name of Elastic IP.
- `remark` - The remarks of Elastic IP.
- `status` - Elastic IP status. Possible values are: `used as in use`, `free as available` and `freeze as associating`.
- `tag` - (Optional) A mapping of tags to assign to VPC, which contains at most 63 characters and only support Chinese, English, numbers, '-', '_', and '.'. If it is not filled in or a empty string is filled in, then default tag will be assigned.

(Default: Default).

The attribute (ip_set) support the following:

- internet_type - Type of Elastic IP routes.
- ip - Elastic IP address

ucloud_images

This data source provides a list of available image resources according to their availability zone, image ID and other fields.

Example Usage

```
data "ucloud_images" "example" {
  availability_zone = "cn-bj2-02"
  image_type       = "base"
  name_regex       = "^CentOS 7.[1-2] 64"
}

output "first" {
  value = "${data.ucloud_images.example.images.0.id}"
}
```

Argument Reference

The following arguments are supported:

- `availability_zone` - (Optional) Availability zone where instances are located. You may refer to [list of availability zone](https://docs.ucloud.cn/api/summary/regionlist) (<https://docs.ucloud.cn/api/summary/regionlist>)
- `image_id` - (Optional) The ID of image.
- `name_regex` - (Optional) A regex string to filter resulting images by name. (Such as: `^CentOS 7.[1-2] 64` means CentOS 7.1 of 64-bit operating system or CentOS 7.2 of 64-bit operating system, `"Ubuntu 16.04 64"` means Ubuntu 16.04 of 64-bit operating system).
- `image_type` - (Optional) The type of image. Possible values are: `base` as standard image, `business` as owned by market place, and `custom` as custom-image, all the image types will be retrieved by default.
- `os_type` - (Optional) The type of OS. Possible values are: `linux` and `windows`, all the OS types will be retrieved by default.
- `output_file` - (Optional) File name where to save data source results (after running `terraform plan`).

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `images` - `images` is a nested type which documented below.
- `total_count` - Total number of image that satisfy the condition.

The attribute (`images`) support the following:

- `create_time` - The time of creation for EIP, formatted in RFC3339 time string.
- `features` - To identify if any particular feature belongs to the instance, the value is `NetEnhanced` as I/O enhanced

instance for now.

- `description` - The description of image if any.
- `id` - The ID of image.
- `name` - The name of image.
- `size` - The size of image.
- `type` - The type of image.
- `os_name` - The name of OS.
- `os_type` - The type of OS.
- `status` - The status of image. Possible values are Available, Making and Unavailable.

ucloud_projects

This data source provides a list of projects owned by user with finance permission.

Example Usage

```
data "ucloud_projects" "example" {
  is_finance = false
}

output "first" {
  value = "${data.ucloud_instances.example.projects.0.id}"
}
```

Argument Reference

The following arguments are supported:

- `is_finance` - (Optional) To identify if the current account is granted with financial permission.
- `output_file` - (Optional) File name where to save data source results (after running `terraform plan`).

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `projects` - It is a nested type which documented below.
- `total_count` - Total number of project that satisfy the condition.

The attribute (`projects`) support the following:

- `create_time` - The time of creation for instance, formatted in RFC3339 time string.
- `id` - The ID of project defined.
- `member_count` - The number of members belongs to the defined project.
- `name` - The name of the defined project.
- `parent_id` - The ID of the parent project where the sub project belongs to.
- `parent_name` - The name of the parent project where the sub project belongs to.
- `resource_count` - The number of the resource instance belong/s to the defined project.

ucloud_zones

This data source provides a list of available zones in the current region.

Example Usage

```
data "ucloud_zones" "example" {}

output "first" {
  value = "${data.ucloud_instances.example.zones.0.id}"
}
```

Argument Reference

The following arguments are supported:

- `output_file` - (Optional) File name where to save data source results (after running `terraform plan`).

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `zones` - Zones is a nested type which documented below.

The attribute (`zones`) support the following:

- `id` - Availability zone where instances are located (such as: `cn-bj-02`). You may refer to list of availability zone (<https://docs.ucloud.cn/api/summary/regionlist>)

ucloud_disk

Provides a Cloud Disk resource.

Example Usage

```
resource "ucloud_disk" "example" {
  availability_zone = "cn-bj2-02"
  name             = "tf-example-disk"
  disk_size        = 10
}
```

Argument Reference

The following arguments are supported:

- `availability_zone` - (Required) The Zone to create the disk in.
- `disk_size` - (Required) Purchase the size of disk in GB. 1-8000 for a cloud disk, 1-4000 for SSD cloud disk.
- `name` - (Optional) The name of disk, should have 6-63 characters and only support Chinese, English, numbers, '-', '_'. If not specified, terraform will autogenerate a name beginning with `tf-disk`.
- `disk_type` - (Optional) The type of disk. Possible values are: `data_disk` as cloud disk, `ssd_data_disk` as ssd cloud disk. (Default: `data_disk`).
- `charge_type` - (Optional) Charge type of disk. Possible values are: `year` as pay by year, `month` as pay by month, `dynamic` as pay by hour. (Default: `month`).
- `duration` - (Optional) The duration that you will buy the resource. (Default: 1). It is not required when `dynamic` (pay by hour), the value is 0 when `month` (pay by month) and the disk will be valid till the last day of that month.
- `tag` - (Optional) A mapping of tags to assign to VPC, which contains at most 63 characters and only support Chinese, English, numbers, '-', '_', and '.'. If it is not filled in or a empty string is filled in, then default tag will be assigned. (Default: `Default`).

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `create_time` - The time of creation of disk, formatted in RFC3339 time string.
- `expire_time` - The expiration time of disk, formatted in RFC3339 time string.
- `status` - The status of disk. Possible values are: `Available`, `InUse`, `Detaching`, `Initializing`, `Failed`, `Cloning`, `Restoring`, `RestoreFailed`.

ucloud_disk_attachment

Provides a Cloud Disk Attachment resource for attaching Cloud Disk to UHost Instance.

Example Usage

```
# Query availability zone
data "ucloud_zones" "default" {}

# Query image
data "ucloud_images" "default" {
  availability_zone = "${data.ucloud_zones.default.zones.0.id}"
  name_regex       = "^CentOS 7.[1-2] 64"
  image_type       = "base"
}

# Create security group
resource "ucloud_security_group" "default" {
  name = "tf-example-disk"
  tag  = "tf-example"

  # allow all access from WAN
  rules {
    port_range = "1-65535"
    protocol   = "tcp"
    cidr_block = "0.0.0.0/0"
    policy     = "accept"
  }
}

# Create security group
resource "ucloud_disk" "default" {
  availability_zone = "${data.ucloud_zones.default.zones.0.id}"
  name              = "tf-example-disk"
  disk_size         = 10
}

# Create a web server
resource "ucloud_instance" "web" {
  availability_zone = "${data.ucloud_zones.default.zones.0.id}"
  instance_type     = "n-standard-1"

  image_id      = "${data.ucloud_images.default.images.0.id}"
  root_password = "${var.instance_password}"

  # this security group allows all access from WAN
  security_group = "${ucloud_security_group.default.id}"

  name = "tf-example-disk"
  tag  = "tf-example"
}

# attach disk to instance
resource "ucloud_disk_attachment" "default" {
  availability_zone = "${data.ucloud_zones.default.zones.0.id}"
  disk_id           = "${ucloud_disk.default.id}"
  instance_id       = "${ucloud_instance.web.id}"
}
```

Argument Reference

The following arguments are supported:

- `availability_zone` - (Required) The Zone to attach the disk in.
- `instance_id` - (Required) The ID of host instance.
- `disk_id` - (Required) The ID of disk that needs to be attached

ucloud_eip

Provides an Elastic IP resource.

Example Usage

```
resource "ucloud_eip" "example" {
  bandwidth      = 2
  charge_mode    = "bandwidth"
  name           = "tf-example-eip"
  tag            = "tf-example"
  internet_type  = "bgp"
}
```

Argument Reference

The following arguments are supported:

- `internet_type` - (Required) Type of Elastic IP routes. Possible values are: `international` as international BGP IP and `bgp` as china BGP IP.
- `bandwidth` - (Optional) Maximum bandwidth to the elastic public network, measured in Mbps (Mega bit per second). the ranges for bandwidth are: 1-200 for pay by traffic, 1-800 for pay by bandwidth. (Default: 1).
- `duration` - (Optional) The duration that you will buy the resource. (Default: 1). It is not required when `dynamic` (pay by hour), the value is 0 when `month`(pay by month) and the instance will be valid till the last day of that month.
- `charge_mode` -(Optional) Elastic IP charge mode. Possible values are: `traffic` as pay by traffic, `bandwidth` as pay by bandwidth. (Default: `bandwidth`).
- `charge_type` - (Optional) Elastic IP charge type. Possible values are: `year` as pay by year, `month` as pay by month, `dynamic` as pay by hour (specific permission required). (Default: `month`).
- `name` - (Optional) The name of the EIP, which contains 1-63 characters and only support Chinese, English, numbers, '-', '_', '!'. If not specified, terraform will autogenerate a name beginning with `tf-eip`.
- `remark` - (Optional) The remarks of the EIP. (Default: "").
- `tag` - (Optional) A mapping of tags to assign to VPC, which contains at most 63 characters and only support Chinese, English, numbers, '-', '_', and '!'. If it is not filled in or a empty string is filled in, then default tag will be assigned. (Default: `Default`).

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `create_time` - The time of creation for EIP, formatted in RFC3339 time string.
- `expire_time` - The expiration time for EIP, formatted in RFC3339 time string.

- `ip_set` - It is a nested type which documented below.
- `resource` - It is a nested type which documented below.
- `status` - EIP status. Possible values are: `used` as in use, `free` as available and `freeze` as associating.

The attribute (`ip_set`) support the following:

- `internet_type` - Type of Elastic IP routes.

The attribute (`resource`) support the following:

- `id` - The ID of the resource with EIP attached.
- `type` - The type of resource with EIP attached. Possible values are `instance` as instance, `vrouter` as virtual router, `lb` as load balancer.

ucloud_eip_association

Provides an EIP Association resource for associating Elastic IP to UHost Instance, Load Balancer, etc.

Example Usage

```
# Query availability zone
data "ucloud_zones" "default" {}

# Query image
data "ucloud_images" "default" {
  availability_zone = "${data.ucloud_zones.default.zones.0.id}"
  name_regex       = "^CentOS 7.[1-2] 64"
  image_type       = "base"
}

# Create security group
resource "ucloud_security_group" "default" {
  name = "tf-example-eip"
  tag  = "tf-example"

  rules {
    port_range = "80"
    protocol   = "tcp"
    cidr_block = "192.168.0.0/16"
    policy     = "accept"
  }
}

# Create an eip
resource "ucloud_eip" "default" {
  bandwidth      = 2
  charge_mode    = "bandwidth"
  name           = "tf-example-eip"
  tag            = "tf-example"
  internet_type  = "bgp"
}

# Create a web server
resource "ucloud_instance" "web" {
  instance_type      = "n-standard-1"
  availability_zone   = "${data.ucloud_zones.default.zones.0.id}"
  image_id           = "${data.ucloud_images.default.images.0.id}"

  data_disk_size = 50
  root_password  = "${var.instance_password}"
  security_group = "${ucloud_security_group.default.id}"

  name = "tf-example-eip"
  tag  = "tf-example"
}

# Bind eip to instance
resource "ucloud_eip_association" "default" {
  resource_type = "instance"
  resource_id   = "${ucloud_instance.web.id}"
  eip_id        = "${ucloud_eip.default.id}"
}
```

Argument Reference

The following arguments are supported:

- `eip_id` - (Required) The ID of EIP.
- `resource_id` - (Required) The ID of resource with EIP attached.
- `resource_type` - (Required) The type of resource with EIP attached. The current possible values are `instance` as instance, `lb` as load balancer.

ucloud_instance

Provides an UHost Instance resource.

Example Usage

```

resource "ucloud_security_group" "default" {
  name = "tf-example-instance"
  tag   = "tf-example"

  # http access from LAN
  rules {
    port_range = "80"
    protocol   = "tcp"
    cidr_block = "192.168.0.0/16"
    policy     = "accept"
  }

  # https access from LAN
  rules {
    port_range = "443"
    protocol   = "tcp"
    cidr_block = "192.168.0.0/16"
    policy     = "accept"
  }
}

resource "ucloud_vpc" "default" {
  name = "tf-example-instance"
  tag   = "tf-example"

  # vpc network
  cidr_blocks = ["192.168.0.0/16"]
}

resource "ucloud_subnet" "default" {
  name = "tf-example-instance"
  tag   = "tf-example"

  # subnet's network must be contained by vpc network
  # and a subnet must have least 8 ip addresses in it (netmask < 30).
  cidr_block = "192.168.1.0/24"
  vpc_id     = "${ucloud_vpc.default.id}"
}

resource "ucloud_instance" "web" {
  name           = "tf-example-instance"
  tag            = "tf-example"
  availability_zone = "cn-bj2-02"
  image_id       = "uimage-of3pac"
  instance_type  = "n-standard-1"

  # use cloud disk as data disk
  data_disk_size   = 50
  data_disk_type   = "local_normal"
  root_password    = "wA1234567"

  # we will put all the instances into same vpc and subnet,
  # so they can communicate with each other.
  vpc_id   = "${ucloud_vpc.default.id}"
  subnet_id = "${ucloud_subnet.default.id}"

  # this security group to allow http and https access
  security_group = "${ucloud_security_group.default.id}"
}

```

Argument Reference

The following arguments are supported:

- `availability_zone` - (Required) Availability zone where instance is located. such as: `cn-bj-02`. You may refer to list of availability zone (<https://docs.ucloud.cn/api/summary/regionlist>)
- `image_id` - (Required) The ID for the image to use for the instance.
- `root_password` - (Required) The password for the instance, which contains 8-30 characters, and at least 3 items of capital letters, lower case letters, numbers and special characters. The special characters include ``()~!@#$%^&*~+=_!{}[];:'<>.,?/`. Note: When it is changed, the instance will reboot to make the change take effect.
- `instance_type` - (Required) The type of instance. There are two types, one is Customized: `n-customized-CPU-Memory`(eg:`n-customized-1-3`), the other is UCloud provider defined: `n-Type-CPU`(eg:`n-highcpu-2`). Thereinto, Type can be `highcpu`, `basic`, `standard`, `highmem` which represent the ratio of CPU and memory respectively (1:1, 1:2, 1:4, 1:8). In addition, range of CPU in core: 1-32, range of memory in MB: 1-256. When it is changed, the instance will reboot to make the change take effect.
- `boot_disk_size` - (Optional) The size of the boot disk, measured in GB (GigaByte). Range: 20-100. The value set of disk size must be larger or equal to 20(default: 20) for Linux and 40 (default: 40) for Windows. The responsive time is a bit longer if the value set is larger than default for local boot disk, and further settings may be required on host instance if the value set is larger than default for cloud boot disk. The disk volume adjustment must be a multiple of 10 GB. When it is changed, the instance will reboot to make the change take effect. In addition, any reduction of boot disk size is not supported.
- `boot_disk_type` - (Optional) The type of boot disk. Possible values are: `local_normal` and `local_ssd` for local boot disk, `cloud_normal` and `cloud_ssd` for cloud boot disk. (Default: `local_normal`). The `local_ssd`, `cloud_normal` and `cloud_ssd` are not supported in all regions as boot disk type, please proceed to UCloud console for more details.
- `data_disk_type` - (Optional) The type of local data disk. Possible values are: `local_normal` and `local_ssd` for local data disk. (Default: `local_normal`). The `local_ssd` is not supported in all regions as disk type, please proceed to UCloud console for more details.
- `data_disk_size` - (Optional) The size of data disk, measured in GB (GigaByte), range: 0-8000 (Default: 20), 0-8000 for cloud disk, 0-2000 for local sata disk and 100-1000 for local ssd disk (all the GPU type instances are included). The volume adjustment must be a multiple of 10 GB. When it is changed, the instance will reboot to make the change take effect. In addition, any reduction of data disk size is not supported.
- `charge_type` - (Optional) The charge type of instance, possible values are: `year`, `month` and `dynamic` as pay by hour (specific permission required). (Default: `month`).
- `duration` - (Optional) The duration that you will buy the instance (Default: 1). The value is 0 when pay by month and the instance will be valid till the last day of that month. It is not required when `dynamic` (pay by hour).
- `name` - (Optional) The name of instance, which contains 1-63 characters and only support Chinese, English, numbers, '-', '_', '.', ':'. If not specified, terraform will autogenerate a name beginning with `tf-instance`.
- `remark` - (Optional) The remarks of instance. (Default: "").
- `security_group` - (Optional) The ID of the associated security group.
- `subnet_id` - (Optional) The ID of subnet.
- `tag` - (Optional) A mapping of tags to assign to VPC, which contains at most 63 characters and only support Chinese,

English, numbers, '-', '_', and '.'. If it is not filled in or a empty string is filled in, then default tag will be assigned.
(Default: Default).

- `vpc_id` - (Optional) The ID of VPC linked to the instance.

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `auto_renew` - Whether to renew an instance automatically or not.
- `cpu` - The number of cores of virtual CPU, measured in core.
- `memory` - The size of memory, measured in MB (Megabyte).
- `create_time` - The time of creation for instance, formatted in RFC3339 time string.
- `expire_time` - The expiration time for instance, formatted in RFC3339 time string.
- `status` - Instance current status. Possible values are `Initializing`, `starting`, `Running`, `Stopping`, `Stopped`, `Install Fail`, `ResizeFail` and `Rebooting`.
- `ip_set` - It is a nested type which documented below.
- `disk_set` - It is a nested type which documented below.

The attribute (`disk_set`) supports the following:

- `id` - The ID of disk.
- `size` - The size of disk, measured in GB (Gigabyte).
- `type` - The type of disk.
- `is_boot` - Specifies whether boot disk or not.

The attribute (`ip_set`) supports the following:

- `internet_type` - Type of Elastic IP routes. Possible values are: `International` as international BGP IP, `BGP as china` BGP IP and `Private` as private IP.
- `ip` - Elastic IP address.

ucloud_lb

Provides a Load Balancer resource.

Example Usage

```
resource "ucloud_lb" "web" {  
  name = "tf-example-lb"  
  tag   = "tf-example"  
}
```

Argument Reference

The following arguments are supported:

- `internal` - (Optional) Indicate whether the load balancer is intranet.
- `charge_type` - (Optional) Charge type of load balancer. Possible values are: `year` as pay by year, `month` as pay by month, `dynamic` as pay by hour (specific permission required). (Default: `month`).
- `name` - (Optional) The name of the load balancer. If not specified, terraform will autogenerate a name beginning with `tf-lb`.
- `vpc_id` - (Optional) The ID of the VPC linked to the Load Balancers, This argumnet is not required if default VPC.
- `subnet_id` - (Optional) The ID of subnet that intrant load balancer belongs to. This argumnet is not required if default subnet.
- `tag` - (Optional) A mapping of tags to assign to VPC, which contains at most 63 characters and only support Chinese, English, numbers, '-', '_', and '.'. If it is not filled in or a empty string is filled in, then default tag will be assigned. (Default: `Default`).
- `remark` - (Optional) The remarks of the load balancer. (Default: is `"`).

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `create_time` - The time of creation for load balancer, formatted in RFC3339 time string.
- `expire_time` - The expiration time for load balancer, formatted in RFC3339 time string.
- `ip_set` - It is a nested type which documented below.
- `private_ip` - The IP address of intranet IP. It is `"` if `internal` is `false`.

The attribute (`ip_set`) support the following:

- `internet_type` - Type of Elastic IP routes.

- ip - Elastic IP address.

ucloud_lb_attachment

Provides a Load Balancer Attachment resource for attaching Load Balancer to UHost Instance, etc.

Example Usage

```
resource "ucloud_lb" "web" {
  name = "tf-example-lb"
  tag  = "tf-example"
}

resource "ucloud_lb_listener" "default" {
  load_balancer_id = "${ucloud_lb.web.id}"
  protocol         = "https"
}

resource "ucloud_security_group" "default" {
  name = "tf-example-eip"
  tag  = "tf-example"

  rules {
    port_range = "80"
    protocol   = "tcp"
    cidr_block = "192.168.0.0/16"
    policy     = "accept"
  }
}

resource "ucloud_instance" "web" {
  instance_type      = "n-standard-1"
  availability_zone  = "cn-bj2-02"

  root_password      = "wA1234567"
  image_id            = "uimage-of3pac"
  security_group     = "${ucloud_security_group.default.id}"

  name               = "tf-example-lb"
  tag                = "tf-example"
}

resource "ucloud_lb_attachment" "example" {
  load_balancer_id = "${ucloud_lb.web.id}"
  listener_id      = "${ucloud_lb_listener.default.id}"
  resource_type    = "instance"
  resource_id      = "${ucloud_instance.web.id}"
  port             = 80
}
```

Argument Reference

The following arguments are supported:

- `load_balancer_id` - (Required) The ID of load balancer instance.
- `listener_id` - (Required) The ID of listener servers.

- `resource_type` - (Required) The types of backend servers. The current possible values are: `instance` as Elastic computing host.
- `resource_id` - (Required) The ID of backend servers.
- `port` - (Optional) Port opened on the backend server to receive requests, range: 1-65535, (Default: 80).

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `private_ip` - The private ip address for backend servers.
- `status` - The status of backend servers. Possible values are: `normalRunning`, `exceptionRunning`.

ucloud_lb_listener

Provides a Load Balancer Listener resource.

Example Usage

```
resource "ucloud_lb" "web" {
  name = "tf-example-lb"
  tag  = "tf-example"
}

resource "ucloud_lb_listener" "example" {
  load_balancer_id = "${ucloud_lb.web.id}"
  protocol         = "https"
}
```

Argument Reference

The following arguments are supported:

- `load_balancer_id` - (Required) The ID of load balancer instance.
- `protocol` - (Required) Listener protocol. Possible values: `http`, `https` if `listen_type` is `request_proxy`, `tcp` and `udp` if `listen_type` is `packets_transmit`.
- `name` - (Optional) The name of the listener. (Default: `Listener`).
- `listen_type` - (Optional) The type of listener. Possible values are `request_proxy` and `packets_transmit`. (Default: `packets_transmit`).
- `port` - (Optional) Port opened on the listeners to receive requests, range: 1-65535. (Default: 80).
- `idle_timeout` - (Optional) Amount of time in seconds to wait for the response for in between two sessions if `listen_type` is `request_proxy`, range: 0-86400. (Default: 60). Amount of time in seconds to wait for one session if `listen_type` is `packets_transmit`, range: 60-900. The session will be closed as soon as no response if it is 0.
- `method` - (Optional) The load balance method in which the listener is. Possible values are: `roundrobin`, `source`, `consistent_hash`, `source_port`, `consistent_hash_port`, `weight_roundrobin` and `leastconn`. (Default: `roundrobin`).
 - The `consistent_hash`, `source_port`, `consistent_hash_port`, `roundrobin`, `source` and `weight_roundrobin` are valid if `listen_type` is `packets_transmit`.
 - The `Roundrobin`, `Source` and `WeightRoundrobin` and `Leastconn` are valid if `listen_type` is `request_proxy`.
- `persistence` - (Optional) Indicate whether the persistence session is enabled, it is invalid if `PersistenceType` is `none`, an auto-generated string will be exported if `persistence_type` is `server_insert`, a custom string will be exported if `persistence_type` is `user_defined`.
- `persistence_type` - (Optional) The type of session persistence of listener. Possible values are: `none` as disabled, `server_insert` as auto-generated string and `user_defined` as custom string. (Default: `none`).

- `health_check_type` - (Optional) Health check method. Possible values are `port` as port checking and `path` as http checking.
- `path` - (Optional) Health check path checking.
- `domain` - (Optional) Health check domain checking.

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `status` - Listener status. Possible values are: `allNormal` for all resource functioning well, `partNormal` for partial resource functioning well and `allException` for all resource functioning exceptional. `

ucloud_lb_rule

Provides a Load Balancer Rule resource to add content forwarding policies for Load Balancer backend resource.

Example Usage

```
resource "ucloud_lb" "web" {
  name = "tf-example-lb"
  tag  = "tf-example"
}

resource "ucloud_lb_listener" "default" {
  load_balancer_id = "${ucloud_lb.web.id}"
  protocol         = "https"
}

resource "ucloud_security_group" "default" {
  name = "tf-example-eip"
  tag  = "tf-example"

  rules {
    port_range = "80"
    protocol   = "tcp"
    cidr_block = "192.168.0.0/16"
    policy     = "accept"
  }
}

resource "ucloud_instance" "web" {
  instance_type      = "n-standard-1"
  availability_zone  = "cn-bj2-02"

  root_password      = "wA1234567"
  image_id           = "uimage-of3pac"
  security_group     = "${ucloud_security_group.default.id}"

  name               = "tf-example-lb"
  tag                = "tf-example"
}

resource "ucloud_lb_attachment" "default" {
  load_balancer_id = "${ucloud_lb.web.id}"
  listener_id      = "${ucloud_lb_listener.default.id}"
  resource_type    = "instance"
  resource_id      = "${ucloud_instance.web.id}"
  port             = 80
}

resource "ucloud_lb_rule" "example" {
  load_balancer_id = "${ucloud_lb.web.id}"
  listener_id      = "${ucloud_lb_listener.default.id}"
  backend_ids      = ["${ucloud_lb_attachment.default.id}"]
  domain           = "www.ucloud.cn"
}
```

Argument Reference

The following arguments are supported:

- `load_balancer_id` - (Required) The ID of the load balancer which requires the rule.
- `listener_id` - (Required) The ID of the listener which requires the rule.
- `backend_ids` - (Required) The IDs of the backend servers where rule applies, this argument is populated base on the `backend_id` responded from `lb_attachment create`.
- `path` - (Optional) The path of Content forward matching fields. `path` and `domain` cannot coexist. `path` and `domain` must be filled in one.
- `domain` - (Optional) The domain of content forward matching fields. `path` and `domain` cannot coexist. `path` and `domain` must be filled in one.

ucloud_security_group

Provides a Security Group resource.

Example Usage

```
resource "ucloud_security_group" "example" {
  name = "tf-example-instance"
  tag  = "tf-example"

  # http access from LAN
  rules {
    port_range = "80"
    protocol   = "tcp"
    cidr_block = "192.168.0.0/16"
    policy     = "accept"
  }

  # https access from LAN
  rules {
    port_range = "443"
    protocol   = "tcp"
    cidr_block = "192.168.0.0/16"
    policy     = "accept"
  }
}
```

Argument Reference

The following arguments are supported:

- **rules** - (Required) A list of security group rules. Each element contains the following attributes: `protocol`, `port_range`, `cidr_block`, `policy` (possible values are: `accept` and `drop`) and `priority` (possible values are: `high`, `medium` and `low`. (eg: `tcp|22|192.168.1.1/22|drop|low`).
- **name** - (Optional) The name of the security group which contains 1-63 characters and only support Chinese, English, numbers, '-', '_' and '!'. If not specified, terraform will autogenerate a name beginning with `tf-security-group`.
- **remark** - (Optional) The remarks of the security group. (Default: "").
- **tag** - (Optional) A mapping of tags to assign to VPC, which contains at most 63 characters and only support Chinese, English, numbers, '-', '_', and '!'. If it is not filled in or a empty string is filled in, then default tag will be assigned. (Default: `Default`).

The attribute (`rules`) support the following:

- **cidr_block** - The cidr block of source.
- **policy** - Authorization policy. Can be either `accept` or `drop`.
- **port_range** - The range of port numbers, range: 1-65535. (eg: `port` or `port1-port2`).
- **priority** - Rule priority. Can be `high`, `medium`, `low`.

- `protocol` - The protocol. Can be `tcp`, `udp`, `icmp`, `gre`. [## Attributes Reference](#)

In addition to all arguments above, the following attributes are exported:

- `create_time` - The time of creation of security group, formatted in RFC3339 time string.

ucloud_subnet

Provides a Subnet resource under VPC resource.

Example Usage

```
resource "ucloud_vpc" "default" {
  name = "tf-example-vpc"
  tag  = "tf-example"

  # vpc network
  cidr_blocks = ["192.168.0.0/16"]
}

resource "ucloud_subnet" "example" {
  name = "tf-example-subnet"
  tag  = "tf-example"

  # subnet's network must be contained by vpc network
  # and a subnet must have least 8 ip addresses in it (netmask < 30).
  cidr_block = "192.168.1.0/24"
  vpc_id     = "${ucloud_vpc.default.id}"
}
```

Argument Reference

The following arguments are supported:

- `cidr_block` - (Required) The cidr block of the desired subnet, format in "0.0.0.0/0", such as: 192.168.0.0/24.
- `vpc_id` - (Required) The id of the VPC that the desired subnet belongs to.
- `name` - (Optional) The name of the desired subnet. If not specified, terraform will autogenerate a name beginning with `tf-subnet`.
- `remark` - (Optional) The remarks of the subnet. (Default: "").
- `tag` - (Optional) A mapping of tags to assign to VPC, which contains at most 63 characters and only support Chinese, English, numbers, '-', '_', and '.'. If it is not filled in or a empty string is filled in, then default tag will be assigned. (Default: Default).

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `create_time` - The time of creation of subnet, formatted in RFC3339 time string.

ucloud_vpc

Provides a VPC resource.

Example Usage

```
resource "ucloud_vpc" "example" {  
  name = "tf-example-vpc"  
  tag   = "tf-example"  
  
  # vpc network  
  cidr_blocks = ["192.168.0.0/16"]  
}
```

Argument Reference

The following arguments are supported:

- `cidr_blocks` - (Required) The CIDR blocks of VPC.
- `name` - (Optional) The name of VPC. If not specified, terraform will autogenerate a name beginning with `tf-vpc`.
- `tag` - (Optional) A mapping of tags to assign to VPC, which contains at most 63 characters and only support Chinese, English, numbers, '-', '_', and '.'. If it is not filled in or a empty string is filled in, then default tag will be assigned. (Default: `Default`).
- `remark` - (Optional) The remarks of the VPC. (Default: `" "`).

Attributes Reference

In addition to all arguments above, the following attributes are exported:

- `create_time` - The time of creation for VPC, formatted in RFC3339 time string.
- `update_time` - The time whenever there is a change made to VPC, formatted in RFC3339 time string.
- `network_info` - It is a nested type which documented below.

The attribute (`network_info`) support the following:

- `cidr_block` - The CIDR block of the VPC.

ucloud_vpc_peering_connection

Provides an VPC Peering Connection for establishing a connection between multiple VPC.

Example Usage

```
resource "ucloud_vpc" "foo" {
  name      = "tf-example-vpc-01"
  tag       = "tf-example"
  cidr_blocks = ["192.168.0.0/16"]
}

resource "ucloud_vpc" "bar" {
  name      = "tf-example-vpc-02"
  tag       = "tf-example"
  cidr_blocks = ["10.10.0.0/16"]
}

resource "ucloud_vpc_peering_connection" "connection" {
  vpc_id      = "${ucloud_vpc.foo.id}"
  peer_vpc_id = "${ucloud_vpc.bar.id}"
}
```

Argument Reference

The following arguments are supported:

- `vpc_id` - (Required) The short of ID of the requester VPC of the specific VPC Peering Connection to retrieve.
- `peer_vpc_id` - (Required) The short ID of acceptor VPC of the specific VPC Peering Connection to retrieve.
- `peer_project_id` - (Optional) The ID of acceptor project of the specific VPC Peering Connection to retrieve.