

Logentries Provider

The Logentries provider is used to manage Logentries logs and log sets. Logentries provides live log management and analytics. The provider needs to be configured with a Logentries account key before it can be used.

Use the navigation to the left to read about the available resources.

Example Usage

```
# Configure the Logentries provider
provider "logentries" {
  account_key = "${var.logentries_account_key}"
}

# Create a log set
resource "logentries_logset" "host_logs" {
  name = "${var.server}-logs"
}

# Create a log and add it to the log set
resource "logentries_log" "app_log" {
  logset_id = "${logentries_logset.host_logs.id}"
  name      = "myapp-log"
  source    = "token"
}

# Add the log token to a cloud-config that can be used by an
# application to send logs to Logentries
resource "aws_launch_configuration" "app_launch_config" {
  name_prefix   = "myapp-"
  image_id      = "${var.ami}"
  instance_type = "${var.instance_type}"

  user_data = <<EOF
#cloud-config
write_files:
- content: |
    #!/bin/bash -l
    export LOGENTRIES_TOKEN=${logentries_log.app_log.token}
    run-my-app.sh
  path: "/etc/sv/my-app/run"
  permissions: 0500
runcmd:
- ln -s /etc/sv/my-app /etc/service/
EOF

  iam_instance_profile = "${var.instance_profile}"

  lifecycle {
    create_before_destroy = true
  }

  root_block_device {
    volume_type = "gp2"
    volume_size = "100"
  }
}
```

Argument Reference

The following arguments are supported in the `provider` block:

- `account_key` - (Required) The Logentries account key. This can also be specified with the `LOGENTRIES_ACCOUNT_KEY` environment variable. See the Logentries account key documentation (<https://logentries.com/doc/accountkey/>) for more information.

logentries_logset

Use this data source to get information (like ID) of already existing Logentries LogSets.

Example Usage

```
data "logentries_logset" "logset" {  
  name = "Example_Logset"  
}  
  
output "example_logset_id" {  
  value = "data.logentries_logset.logset.id"  
}
```

Attributes Reference

- name - Name of the LogSet to query.

logentries_log

Provides a Logentries log resource.

Example Usage

```
# Create a log and add it to the log set
resource "logentries_log" "app_log" {
  logset_id = "${logentries_logset.host_logs.id}"
  name      = "myapp-log"
  source    = "token"
}
```

Argument Reference

The following arguments are supported:

- `logset_id` - (Required) The id of the `logentries_logset` resource.
- `name` - (Required) The name of the log. The name should be short and descriptive. For example, Apache Access, Hadoop Namenode.
- `filename` - (Optional) the filename of the log.
- `retention_period` - (Optional, default `ACCOUNT_DEFAULT`) The retention period (1W, 2W, 1M, 2M, 6M, 1Y, 2Y, UNLIMITED, `ACCOUNT_DEFAULT`)
- `source` - (Optional, default `token`) The log source (`token`, `syslog`, `agent`, `api`). Review the Logentries log inputs documentation (<https://docs.logentries.com/docs/>) for more information.
- `type` - (Optional) The log type. See the Logentries log type documentation (<https://logentries.com/doc/log-types/>) for more information.

Attributes Reference

The following attributes are exported:

- `token` - If the log source is `token`, this value holds the generated log token that is used by logging clients. See the Logentries token-based input documentation (<https://logentries.com/doc/input-token/>) for more information.

logentries_logset

Provides a Logentries logset resource. A logset is a collection of `logentries_log` resources.

Example Usage

```
# Create a log set
resource "logentries_logset" "host_logs" {
  name      = "${var.server}-logs"
  location  = "www.example.com"
}
```

Argument Reference

The following arguments are supported:

- `name` - (Required) The log set name, which should be short and descriptive. For example, `www`, `db1`.
- `location` - (Optional, default "nonlocation") A location is for your convenience only. You can specify a DNS entry such as `web.example.com`, IP address or arbitrary comment.